
Devoir surveillé 3 − Correction

Ce sujet est issu de la troisième partie de l’épreuve de mathématiques ÉCRICOME 2016. Voici
un extrait du rapport du jury sur cet exercice :

Rappelons encore une fois que l’informatique représente une part non négligeable du
programme des deux années de classes préparatoires. Les questions proposées dans les
sujets sont souvent élémentaires et bien rémunérées lors de la correction, ce qui doit
encourager tout candidat à aborder les parties du sujet portant sur les connaissances
en Scilab. Il n’a pas été rare de voir des candidats de niveau plutôt moyen par ailleurs
aborder l’exercice 3 raisonnablement, leur faisant ainsi augmenter considérablement leur
note finale.

Q.1 On veut simuler le tirage d’une boule, blanche ou noire. Il faut donc connaître les proba-
bilités de tirer une boule blanche/noire. On a :

P("tirer un boule blanche") =
nombre de boules blanches

nombre de boules
=

b

b+ n
.

On peut désormais compléter le programme, puisque l’événement "tirer une boule blanche" doit
correspondre à l’événement "le programme renvoie 2", d’après la convention prise dans la définition
des variables aléatoires X et Y .

function res = tirage(b,n)
r = rand()
if r < b/(b+n) then

res = 2
else

res = 1
end

endfunction

En effet, comme la variable rand() suit la loi uniforme sur [0,1],

P(r < b/b+n) =
b

b+ n
= P("tirer une boule blanche").

Q.2 Dans cette question, on simule l’expérience selon la variante considérée. Dans la variante
3, il ne se passe rien, donc il n’est pas nécessaire de la prendre en compte dans le programme. Il faut
ici se souvenir de la convention utilisée : 1 signifie que la boule considérée est noire, 2 signifie que la
boule est blanche.

Remarquons que l’énoncé du concours ne comportait pas assez de lignes à remplir. En effet,
puisque l’énoncé n’utilise pas le mot clé elseif (collé), le else if variante==2... correspond à la
création d’une nouvelle boucle if, qu’il faut donc fermer.

function [x,y] = experience(b,n,c,variante)
x = tirage(b,n) //on tire la première boule elle est noire (x=1) ou blanche (x=2)
if variante == 1 then

if x == 1 then //si on a obtenu une boule noire
n = n + c //on ajoute c boules noires

else //sinon, on a obtenu une boule blanche
b = b + c //on ajoute c boules blanches

end
else if variante == 2 then

if x == 1 then //si on a obtenu une boule noire
b = b + c //on ajoute c boules blanches

else //sinon, on a obtenu une boule blanche
n = n + c //on ajoute c boules noires

end //on ferme la boucle if x==1
end //on ferme la boucle if variante==2

end
y = tirage(b,n)

endfunction

Q.3 On répète N fois l’expérience. Ici, il faut voir loiX, loiY et loiXY comme des tableaux
contenant des compteurs :

— loiX(1) est un compteur du nombre de fois que l’expérience a donné X = 1, c’est-à-dire le
nombre de fois que la première boule tirée a été noire,

— loiX(2) est un compteur du nombre de fois que l’expérience a donné X = 2, c’est-à-dire le
nombre de fois que la première boule tirée a été blanche,

— loiY(1) est un compteur du nombre de fois que l’expérience a donné Y = 1, c’est-à-dire le
nombre de fois que la deuxième boule tirée a été noire,

— loiY(2) est un compteur du nombre de fois que l’expérience a donné Y = 2, c’est-à-dire le
nombre de fois que la deuxième boule tirée a été blanche,

— loiXY(1,1) est un compteur du nombre de fois que l’expérience a donné X = 1 et Y = 1,
— loiXY(1,2) est un compteur du nombre de fois que l’expérience a donné X = 1 et Y = 2,
— loiXY(2,1) est un compteur du nombre de fois que l’expérience a donné X = 2 et Y = 1,
— loiXY(2,2) est un compteur du nombre de fois que l’expérience a donné X = 2 et Y = 2,

function [loiX,loiY,loiXY] = estimation(b,n,c,variante,N)
loiX = [0,0] //avant le début des expériences, les compteurs sont à 0
loiY = [0,0]
loiXY = [0,0;0,0]
for k = 1:N

[x,y] = experience(b,n,c,variante) //le résultat de l’expérience est X=x et Y =y
loiX(x) = loiX(x) + 1 //on a obtenu X=x donc on augmente de 1

//le compteur correspondant
loiY(y) = loiY(y) + 1 //idem
loiXY(x,y) = loiXY(x,y) + 1 //idem

end
loiX = loiX/N //on se ramène à des fréquences
loiY = loiY/N
loiXY = loiXY/N

endfunction

Q.4 Il s’agit d’interpréter les résultats. On a obtenu des estimations des lois de X, de Y et de la
loi du couple (X,Y ). Pour vérifier l’indépendance et l’échangeabilité, il faudrait au contraire raisonner
sur les valeurs exactes. Ici, on va donc devoir avoir une certaine souplesse quant à l’utilisation du
symbole ', tout en gardant en tête que l’on a répété l’expérience 10 000 fois et donc que les lois
obtenues devraient (on l’espère) être assez précises.

Vérification de l’échangeabilité C’est le plus facile. Par définition, X et Y sont échangeables
si

∀(i, j) ∈ {1, 2}2, P({X = i} ∩ {Y = j}) = P({X = j} ∩ {Y = i}).
Cette condition n’impose rien lorsque (i, j) = (1, 1) ou (i, j) = (2, 2). Il suffit donc de vérifier que

P({X = 1} ∩ {Y = 2}) = P({X = 2} ∩ {Y = 1})

soit
loiXY(1,2) ' loiXY(2,1).

— Variante 1 : on a loiXY(1,2) = 0.16785 ' 0.16697 = loiXY(2,1) (on a égalité à 10−3 près).
Donc dans la variante 1, les variables X et Y sont échangeables.

— Variante 2 : on a loiXY(1,2) = 0.33286 6' 0.25031 = loiXY(2,1) (la différence entre les deux
est supérieure à 0.05). Donc dans la variante 2, les variables X et Y ne sont pas échangeables.
On aurait également pu utiliser le résultat préliminaire 2 indiqué au début de l’énoncé, puisque
si X et Y avait été échangeable, on aurait du avoir loiX(1) ' loiY(1), ce qui n’est pas le cas.

— Variante 3 : on a loiXY(1,2) = 0.22098 ' 0.22312 = loiXY(2,1) (la différence entre les
deux est inférieure à 3.10−3). Donc dans la variante 3, les variables X et Y sont échangeables.

Vérification de l’indépendance Par définition, X et Y sont indépendantes si

∀(i, j) ∈ {1, 2}2, P({X = i} ∩ {Y = j}) = P(X = i)× P(Y = j).

On doit donc vérifier que 
loiXY(1,1) ' loiX(1)× loiY(1)
loiXY(1,2) ' loiX(1)× loiY(2)
loiXY(2,1) ' loiX(2)× loiY(1)
loiXY(2,2) ' loiX(2)× loiY(2).

Si l’une de ces "égalités" n’est pas vérifiée, on peut directement conclure que X et Y ne sont pas
indépendantes.

— Variante 1 : on a loiXY(1,1) = 0.49837 alors que

loiX(1)× loiY(1) ' 0.66× 0.66 ' 0.44

donc loiXY(1,1) 6' loiX(1)× loiY(1), donc X et Y ne sont pas indépendantes.

— Variante 2 : on a loiXY(1,1) = 0.33258 alors que

loiX(1)× loiY(1) ' 0.66× 0.58 ' 0.38

donc loiXY(1,1) 6' loiX(1)× loiY(1), donc X et Y ne sont pas indépendantes.

— Variante 3 : dans ce cas, X et Y semblent être indépendantes.

